
Lab 7: Longitudinal Binary Data

Overview

Binary data
1. Inspect the Kenya Morbidity data.  

2. Fit a random intercept model to the data with Proc NLMIXED

a. Expand the model

3. Fit the same model using SAS Proc glimmix

4. Modify the analyses

Use the Kenya Morbidity data for this lab.  We model the probability ij of presence of malaria (Yij = 1) or other disease or morbidity over time in subjects partaking in the Kenya school lunch study.  Our model is 

Yij ~ Bernoulli(ij)


logit(ij) = intercept + age_at_time0 + relmonth*treatment + gender + cos(2relyear) + sin(2relyear) + i

i ~ N(0,D)

This is a logistic random intercept model.  All treatment groups are assumed to start at the same point at time 0, then spread out differently over time.  The sin and cos terms model seasonal cycles in the probability of morbidity.  

Lab Tasks

Download the Kenya morbidity data.  If you are at home and do not have the Glimmix procedure because you have an older version of SAS, you may download it at http://support.sas.com/rnd/app/da/glimmix.html. 
Please change directory names as appropriate for your computer.  

/* Download PROC GLIMMIX  http://support.sas.com/rnd/app/da/glimmix.html*/
%let workdir = C:\Documents and Settings\sph\Desktop\; 

/*or current directory where your morbidity data is saved*/
libname data "&workdir";

Title and some options.  

OPTIONS PAGENO=1 PS=75 LS=80 CENTER NODATE MPRINT;

title1 "Using sine and cosine to model season effects";

1.  Inspect the data

proc contents data=data.morbidity_class;

run;

proc print data=data.morbidity_class;

where ID <5;

run;

TODO: (1) Inspect the data.  How many morbidity variables are in the data?  Which one is malaria?  

TODO: (2) How many subjects, how many observations?  

TODO: (3) There is one variable that is the indicator of a severe morbidity versus a mild or no morbidity, and another that is severe and mild versus no morbidity.  Identify those two variables.  

2. Fit using Proc NLMIXED

Proc nlmixed requires starting values for the parameters, which I have supplied here in the parms statement.  These are not estimates!  They are initial values for the iterative computational algorithm to use.  The key words on the proc line tell SAS how to do the integration of the random effects, and ecov gives the covariance of the estimates.  Unfortunately, nlmixed does not have features that mixed and other sas procs have, consequently we have to explicitly write the fixed effects by hand.  The comment rewrites the model as if we were using proc mixed instead.

The model is a logistic random effects model.  Eta is the linear predictor, a0 – a8 are fixed effects parameters, and b1 is a random intercept.  That b1 is a subject specific parameter is only specified at the random statement.  The parameter p is the inverse logit (expit) of the linear predictor.  P (really, p_{ij}) is the probability that y_{ij}is 1.  The random intercept is assumed distributed normally, with variance d.  There is a separate slope for each treatment group, a gender effect and an age at baseline effect.  Additionally there is a sin and cos term that allow for seasonal variation in probabilities.  

title2 "nlmixed malaria"; 

proc nlmixed data=data.morbidity_class method=gauss noad qfac=11 ecov;

*  

model q15 = age_at_time0 sex relmonth treatment*relmonth sin cos/s ;
      parms a0=-.35 a1=-.16 a2=-.19 a3=-.08 a4=-.08 a5=-.08 





a6=-.08 a7=-.15 a8=.20 d=.75;

      eta = a0 + a1 * age_at_time0 + a2 * boy + 




a3 * relmonth * (1 - tmtcalorie - tmtmilk - tmtmeat) + 




a4 * relmonth * tmtcalorie + 




a5 * relmonth * tmtmilk + a6 * relmonth * tmtmeat + 




a7 * sin + a8 * cos + b1;

      expeta = exp(eta);

      p = expeta/(1+expeta);

      model Q15 ~ binomial(1,p);

      random b1 ~ normal(0,d) subject=id;

*      predict eta out=eta;
      estimate 'calorie-control' a4-a3;

      estimate 'milk-control' a5-a3;

      estimate 'meat-control' a6-a3;

      estimate 'calorie-milk' a4-a5;

      estimate 'calorie-meat' a4-a6;

      estimate 'milk-meat' a5-a6;


  estimate 'age/12 - time' a1/12 - (a3 + a4 + a5 + a6)/4;


  estimate 'age/12 - time' a1/12 - a3;



contrast 'sin=cos=0' a7, a8;



contrast 'treatment' a4-a3, a5-a4, a6-a5;

   run;

 The estimate statements calculate some contrasts of interest.  The contrast statements test for whether we need the sin and cos term in the model and whether the treatment slopes are different.  The two ‘age/12 – time’ effects compare the slope of age to the slopes of time in study in two ways to see if aging has the same effect as being in the study does.  The two estimates are significantly different, indicating that the two effects are different.  

TODO:  (4) Find the estimates of the time slopes of the four treatment groups.  Are these slopes significantly different from zero or not?  Are these slopes significantly different from each other?  

TODO:  (5) Find the contrast statements output and interpret it.  

TODO:  (6) Find the output from the estimate statements and interpret that as well.  

TODO:  (7) What is the variance of the random intercepts?  

3. Fit using PROC GLIMMIX  

The GLIMMIX procedure fits statistical models to data with correlations or nonconstant

variability and where the response is not necessarily normally distributed. These

models are known as generalized linear mixed models (GLMM).  (Glimmix is derived from the earlier glimmix macro.  Glimmix is quite powerful; it can fit multiple outcome hierarchical models, models with multiple outcomes of several types.  For example, you can fit a bivariate longitudinal model with one outcome a count, and another outcome a binary outcome.)
   title2 "PROC GLIMMIX for malaria"; 

proc glimmix data=data.morbidity_class noitprint noclprint order=internal method=MMPL ic=PQ;



where q15 ne .;

  

class id SCHOOL sex treatment vn;

  

model q15 =age_at_time0  sex  relmonth  treatment*relmonth sin cos/s dist=binomial link=logit ;



random intercept / sub=id(SCHOOL) type=un;


estimate 'relMonth slope cal  - meat' relMonth*treatment 1 -1  0  0;

      estimate 'relMonth slope cal  - milk' relMonth*treatment 1  0 -1  0;


estimate 'relMonth slope cal  - cont' relMonth*treatment 1  0  0 -1;


estimate 'relMonth slope meat - milk' relMonth*treatment 0  1 -1  0;

 
estimate 'relMonth slope meat - cont' relMonth*treatment 0  1  0 -1;


estimate 'relMonth slope milk - cont' relMonth*treatment 0  0  1 -1;


contrast 'sin=cos=0' sin 1, cos 1;

run;

method = RMPL stands for residual likelihood (“R”) or the first M in MMPL stands for maximum likelihood
Second M is for “M”arginal expansions 

Pseudo-likelihood methods for generalized linear mixed models can be cast in terms of Taylor series expansions (linearizations) of the GLMM. The expansion locus of the expansion is either the vector of random effects solutions (“S”) or the mean of the random effects (“M”). The

expansions are also referred to as the “S”ubject-specific and “M”arginal expansions. These are two different methods of approximating the marginal likelihood. 
“PL” are pseudo-likelihood techniques

IC=PQ option requests that the penalties include the number of fixed effects parameters, when estimation in models with random effects is based on a residual (restricted) likelihood.

*/

The next 4 todo steps here for the glimmix run are the same as for the previous nlmixed run.  

TODO:  (4) Find the estimates of the time slopes of the four treatment groups.  Are these slopes significantly different from zero or not?  Are these slopes significantly different from each other?  

TODO:  (5) Find the output of the contrast statement and interpret it.  (6a) The nlmixed code had an extra contrast statement that is not needed here.  Find the equivalent output from glimmix and interpret it.  
TODO:  (6) Find the output from the estimate statements and interpret that as well.  

TODO:  (7) What is the variance of the random intercepts?  

TODO:  (8) How does the output in the two runs differ?  Are the differences important?  

4. Modify the analyses.  

For both analyses.  

TODO: (9) Calculate the probability of malaria at time 0 and time relmonth=12. Plot the probability of malaria over time.  

TODO:  (10) These analyses both force the intercepts for the four treatment groups to be the same.  (a) Modify the glimmix program to have separate intercepts for the four groups.  This requires basically a one-word change to the code.  (b) (harder or at least more annoying) Modify the nlmixed program to have separate intercepts for the four groups.  You will need to create 3 new ‘a’ parameters, and also give them starting values in the parms statement.  

TODO: (11) Change the response variable to severe versus mild/no morbidity in both analyses.  Are the results similar to malaria or different?  If different, how do the results differ?
Note: In the Methods= keyword in Glimmix, not all choices always lead to convergence. Generally I try out all of Methods = RSPL | MSPL | RMPL | MMPL and see which method leads to convergence and use that. In this example, MMPL converges, and if you change maxopt=100, then RMPL will also converge. Extra todo: is the fit much different between these two choices? 

